标题：抽筋宏
抽筋宏的作用、优劣及分类
早期猎人的抽筋宏是为了能够比较完美的安排稳固射击和自动射击两个技能而诞生的。
现阶段随着稳固的修改这种作用已经不再存在，主要作用转变为了安排技能流程，使得战斗更加简单。
现阶段的抽筋宏都是使用的顺序施放（施放序列/castsequence命令） ，不可避免的有一个共同的问题，那就是施放途中如果打断宏使用其他技能或物品，都会影响宏的正常运行，导致卡宏，直到宏重置（reset=）。这一点是无法避免的，所以请尽量在宏快要重置或者重置时使用其他技能或物品。
同时，现阶段的抽筋宏无论多么完美都不如严格的手动操作DPS高，只是可以节省精力以及避免人为失误。
网易版国服3.13射击完美抽筋宏
有一拖二宏和单宏两个版本，各有优点，请自行选用。
一拖二版宏宏一
#showtooltip 稳固射击
/施放 !自动射击
/施放 [target=pettarget,exists]杀戮命令
/castsequence reset=10 奇美拉射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,
/click MultiBarBottomRightButton11
宏二(放在底部右边动作栏倒数第二个位置
/castsequence reset=10 瞄准射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击
/click MultiBarBottomRightButton12
宏三 (放在底部右边动作栏最后一个位置
/castsequence reset=6 奥术射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击,自动射击
/施放 稳固射击
/cast 沉默射击
  
开始手动上个毒蛇然后一直抽宏一即可
宏二和宏三的位置如图
单宏版#showtooltips 奇美拉射击
/startattack
/petattack
/castsequence reset=9 奇美拉射击,毒蛇钉刺,奥术射击,稳固射击,稳固射击,稳固射击,稳固射击,稳固射击,稳固射击，稳固射击
/cast [target=pettarget,exists]杀戮命令
/script UIErrorsFrame:Clear()
使用前请自己先施放一个毒蛇钉刺。请自己调整稳固射击的数量以保证奇美拉射击能持续刷新毒蛇钉刺。
因为本人是使用奥术而非瞄准的，所以如果需要使用瞄准、沉默射击，请自己将奥术射击替换为瞄准射击（注意要配合瞄准射击雕文使用），沉默射击推荐自行在宏即将重置是使用。
3.0X版本 射击天赋抽筋宏
攻击模式 
毒蛇-奇美拉-稳固-稳固-稳固-稳固-稳固-稳固-稳固-奇美拉-稳固-稳固-稳固-稳固-稳固-稳固-稳固-奇美拉-........ 
#showtooltips 奇美拉射击 
/console Sound_EnableSFX 0 
/castsequence reset=3 奇美拉射击,稳固射击,稳固射击,稳固射击,稳固射击,稳固射击,稳固射击 
/console Sound_EnableSFX 1 
/脚本 UIErrorsFrame:Clear()
WLK80级兽王猎人极简完美抽筋宏
/cast !自动射击 
/cast 击杀射击 
/cast 稳固射击 
/cast 击杀命令 
/petattack
意思为
施放 自动射击 
施放 击杀射击 
施放 稳固射击 
施放 击杀命令 
BB攻击此目标
使用AutoHotKey 让猎人的手指不再抽筋
WLK可用
在美服新猎人宏的帖子里面, 曾经提到说这个宏需要每秒击发10次才能达到最好的效果, 当然如果你用手指点的话是绝对不可能达到每秒10次的速度的, 所以在哪个帖子里面作者提到了一个软件, 那就是AutoHotKey, 用他来达到理想的每秒10次的速度, 使LR的手指不再抽筋!此工具和按键精灵同理，但比按键精灵更省系统资源，所以推荐使用。
现在来说下安装, 及使用的步骤.
1、首先下载, 并安装这个软件
地址: http://www.autohotkey.com/download/AutoHotkey104705_Install.exe
http://www.autohotkey.com/download/AutoHotkey104705_Install.exe 
2、安装完后, 在第一次启动的时候会有以下的选向弹出
选Yes
3、这时, 要把文件内的所有内容删掉, 并且把以下内容拷贝到文件内
#Persistent
#NoEnv
v_Enable=0
$e::
Loop
{
if not GetKeyState("e", "P") 
break
Send 
Sleep, 100
}
return
然后关闭并且保存文件 wow.ahk
把你的抽筋宏　放到　e号格子位上
此脚本意思是，按住e,输出宏开始，松开e,宏结束。
这样一来不用像平时一样狂按抽筋，按都手到酸了。
另：如想换其它键,只需要将上面的 e 改为你想改的键，再将宏放到对应键上即可。
历史2.4x版前抽筋宏
1：1抽筋宏
1：1抽筋宏使稳固射击在两个自动射击的间隙中施放出来，使整个攻击过程呈现出1稳固-1自动的射击循环（这个循环是兽王猎人最合适的射击循环），并且在能够使用杀戮命令的时候同时施放杀戮命令。在这个宏形成的循环中，稳固与自动的比例是1：1，因此叫做1：1抽筋宏。
2：3抽筋宏
在1：1抽筋宏中，依照稳固-自动的顺序进行技能的施放存在一个不足之处：而对于较慢速度的武器，按照1：1进行稳固-自动会浪费一些时间，然而对于兽王来说，也不大可能使用射击天赋的射击循环（对于射击所用的稳固-多重-自动-稳固-奥术-自动循环来说，由于兽王本身天赋的20%提速，速度又太快了）。因此兽王有时候会用一些加速装备减少浪费的时间。同时，一种自动调整稳固与自动的宏也出现了，由于使用这个宏基本上使得自动与稳固依照2：3的概率触发，因此被称作2：3宏。经测试，2：3宏和1：1宏相比，输出有一定的提高，但是耗蓝量却和输出的提升不成正比。这使得2：3宏的输出续航能力不强，往往只能在DPS全开的rush阶段或者不缺蓝的情况下（比如队里有暗牧和萨满，目标被上了智慧审判，自己还有智慧祝福的理想回蓝情况）使用。
抽筋宏操作简单，只需要狂按键盘就可以达到很高的输出。缺点是对键盘和手指的损害较大，所以也有一些玩家使用精确计时的宏，精确计时宏的优点是按键频率低，但是对节奏的把握需要较高的要求，否则就会打乱射击循环，反而影响输出
一个1:1抽筋宏的范本
现在魔兽世界已经更新到了2.3.2版本，2.3.2版本解决了原来版本中可能出现的卡自动的现象，同时抽筋宏的写法也有了一些改变，2.3.2版本的抽筋宏的一个范本如下：
#showtooltip 稳固射击
/castsequence reset=3 稳固射击,!自动射击
/cast [target=pettarget,exists] 杀戮命令
/script UIErrorsFrame:Clear();
一个2:3抽筋宏的范本
2：3宏是一个以稳固为重点的宏，它的作用机理是当稳固射击可以施放时，施放稳固射击，在稳固不能释放时施放自动射击。
#showtooltip 稳固射击
/cast 稳固射击
/cast !自动射击
/cast [target=pettarget,exists] 杀戮命令
/script UIErrorsFrame:Clear();
一个完美的精确计时宏的范本
抽筋宏虽然使用方便，但是需要过于频繁的按键，对手指和键盘的损伤较大，因此并不让部分玩家所满意。另一种稳固宏——精确计时宏则更为他们所偏爱。但是精确计时宏对于触发时间的要求比较高，需要紧跟着自动射击后按键，在网络条件不好延迟高的时候可能会影响到输出。但是，在正常情况下，抽筋宏与计时宏的作用与效果是等效的，所不同的只是操作方式。
#showtooltip 稳固射击
/stopcasting
/cast [target=pettarget,exists]杀戮命令
/stopcasting [nochanneling]
/script UIErrorsFrame:Clear()
/cast 稳固射击
一个完美的风蛇抽筋宏：
风蛇抽筋宏是传统抽筋宏的一个衍生，对风蛇的闪电吐息技能进行了优化。
#showtooltip 稳固射击 
/cast !自动射击 
/petautocastoff 闪电吐息 
/cast [target=pettarget, exists] 杀戮命令 
/cast [target=pettarget, exists] 杀戮命令 
/cast [pet:风蛇,target=pettarget]闪电吐息 
/cast 稳固射击 
/script UIErrorsFrame:Clear() 
关于杀戮命令吃掉自动射击的解决办法
提供一个宏 
#show 稳固射击 
/click [target=pettarget,exists] MultiBarBottomLeftButton11 
/petautocastoff 闪电吐息 
/cast !自动射击 
/cast [pet:风蛇,target=pettarget]闪电吐息 
/cast 稳固射击 
/script UIErrorsFrame:Clear() 
MultiBarBottomLeftButton11 
这个的意思是：另一个按键上的技能插在这个宏上，等于1个按键带动2个宏 
LeftButton11是左上方按键11，我放的那个宏，可以根据自己的按键修改 
这里提供另一个宏 
/castsequence reset=4 杀戮命令,!自动射击,!自动射击,!自动射击 
这个宏是我自己编的，可以将reset=4改成6，但是我不确定，我测试下来reset=4效果还是不错的，因为杀戮命令的CD是6S 
把这个宏放在左上方按键11（可以自己改） 
这样就可以在原来的DPS基础上再提升一部分
这个宏的优点
1.在稳固和自动都存在的条件下，优先稳固 
2.在杀戮和自动都存在的条件下，优先自动 
3.相对原来的稳固宏更省蓝，DPS更高